Kunsten at løse ligninger i Maple

© *Knud Nissen & Adept Scientific 2008*

Indledning

Maple har en meget stærk kommando til at løse ligninger symbolsk:

> $solve(Ligning)$
> $solve(ligning, Ukendt)$
> $solve(Ligning, DelUdtryk)$

Den første form forudsætter, at alle de ubekendte fremgår af ligningen. Man kan altså nemt løse en konkret ligning som

> $solve(3x + 5 = 7)$

$$\frac{2}{3} \tag{1.1.1}$$

Hvis man vil løse en symbolsk ligning på denne form, så går det galt

> $solve(a \cdot x = b)$

$$\{a = a, b = a\,x, x = x\} \tag{1.1.2}$$

Maple ved jo ikke, hvad der er den ubekendte. Læg mærke til de krøllede parenteser. Det er Maples måde at håndtere mængder på: Løsningen indeholder 3 elementer (i dette tilfælde ligninger): $a = a$, $x = x$ og $b = a \cdot x$. Det skal forstås som en parametrisering af løsningen, hvor a og x kan antage vilkårlige værdier, mens b er fastlagt som produktet af a og x.
I stedet må man skrive

> $solve(a \cdot x = b, x)$

$$\frac{b}{a} \tag{1.1.3}$$

Man kan også få skrevet løsning ud som en mængde ved fx at tilføje mængdeparenser omkring de ubekendte (det være bekvemt i forskellige sammenhænge):

> $solve(a \cdot x = b, \{x\})$

$$\left\{x = \frac{b}{a}\right\} \tag{1.1.4}$$

Endelig kan man også bruge solve-kommandoen til isolere et vilkårligt udtryk i en ligning, men Maple anbefaler ikke denne brug. Fx

> $solve\left(E_{kin} = \frac{1}{2} m \cdot v^2, v^2\right)$

```
Warning, solving for expressions other than names or
functions is not recommended.
```

$$\frac{2\,E_{kin}}{m} \tag{1.1.5}$$

Ligninger kan navngives, så der senere kan refereres til dem via navnet (her *eq1*):

> $eq1 := x^3 + x^2 - 3x - 2 = 0$

$$eq1 := x^3 + x^2 - 3\,x - 2 = 0 \quad \textbf{(1.1.6)}$$

Løsningerne til ligningen kan også navngives, og det er en uhyre nyttig facilitet, så man senere kan få fat i de enkelte løsninger:

> $L := solve(eq1, x)$

$$L := -2, \frac{1}{2} + \frac{1}{2}\sqrt{5}, \frac{1}{2} - \frac{1}{2}\sqrt{5} \quad \textbf{(1.1.7)}$$

Løsningerne returnes som en liste, og de enkelte elementer kan udtrækkes med $L[\ ]$ eller $L\_$. Fx er (L_2 indtastes som L_2):

> L_2

$$\frac{1}{2} + \frac{1}{2}\sqrt{5} \quad \textbf{(1.1.8)}$$

Hvis du sætter krøllede parenteser om x, får du løsningerne på formen

> $L2 := solve(eq1, \{x\})$

$$L2 := \{x = -2\}, \left\{x = \frac{1}{2} + \frac{1}{2}\sqrt{5}\right\}, \left\{x = \frac{1}{2} - \frac{1}{2}\sqrt{5}\right\} \quad \textbf{(1.1.9)}$$

Dette kan være meget nyttigt, hvis du skal substituere en af løsningerne i et udtryk. Fx kan løsningerne checkes således:

> $subs\left(L2_2, eq1\right)$

$$\left(\frac{1}{2} + \frac{1}{2}\sqrt{5}\right)^3 + \left(\frac{1}{2} + \frac{1}{2}\sqrt{5}\right)^2 - \frac{7}{2} - \frac{3}{2}\sqrt{5} = 0 \quad \textbf{(1.1.10)}$$

> $expand(\textbf{(1.1.10)})$

$$0 = 0 \quad \textbf{(1.1.11)}$$

Sådan kan løsningsmængden begrænses

Hvis du blot er interesseret i den positive løsning i ligningen $x^2 = 2$, kan du føje en begrænsningsulighed til ligningen:

> $L3 := solve(\{x^2 = 2, x > 0\}, x)$

$$L3 := \{x = \sqrt{2}\} \quad \textbf{(1.2.1)}$$

Prøv også at finde de negative løsninger til ligningen $eq1$:

> $solve(\{eq1, x < 0\}, x)$

$$\{x = -2\}, \left\{x = \frac{1}{2} - \frac{1}{2}\sqrt{5}\right\} \quad \textbf{(1.2.2)}$$

Helt så nemt går det ikke altid, fx hvis vi kun er interesserede i en positiv løsning til ligningen:

> $eq2 := x^2 + x = a^2$

$$eq2 := x^2 + x = a^2 \quad \textbf{(1.2.3)}$$

> $solve(eq2, x)$

$$-\frac{1}{2}+\frac{1}{2}\sqrt{1+4a^2}, -\frac{1}{2}-\frac{1}{2}\sqrt{1+4a^2} \tag{1.2.4}$$

> $solve(\{eq2, x>0\}, x)$

$$\begin{cases} \left[\left\{x=-\frac{1}{2}+\frac{1}{2}\sqrt{1+4a^2}\right\}\right] & a \neq 0 \\ [\,] & \mathit{otherwise} \end{cases} \tag{1.2.5}$$

For at få en positiv løsning ses det, at $a \neq 0$. Dette kan vi klare med en *assume*-kommando:

> $assume(a \neq 0)$

> $solve(\{eq2, x>0\}, x)$

$$\left\{x=-\frac{1}{2}+\frac{1}{2}\sqrt{1+4a\sim^2}\right\} \tag{1.2.6}$$

Der er så kommet en $\sim$ efter a. Det betyder, at der er gjort antagelser om a.

Sådan undgår du komplekse løsninger

Løsning af ligningen $x^3=1$ vil give 3 løsninger, hvoraf to er komplekse:

> $solve(x^3=1, x)$

$$1, -\frac{1}{2}+\frac{1}{2}\mathrm{I}\sqrt{3}, -\frac{1}{2}-\frac{1}{2}\mathrm{I}\sqrt{3} \tag{1.3.1}$$

Som standard løser solve-kommandoen over de komplekse tal. Dette kan du undgå ved at indlæse pakken *RealDomain*. Som standard arbejder Maple over de komplekse tal - dette ændres til de reelle tal ved indlæsning af *RealDomain*-pakken:

> $with(RealDomain):$

> $solve(x^3=1, x)$

$$1 \tag{1.3.2}$$

> $unwith(RealDomain)$

Sådan får du alle løsninger med

Ved løsning af ligningen $\sin(x)=\frac{1}{2}$, som har uendelig mange løsninger, returnerer Maple blot en enkelt:

> $solve\left(\sin(x)=\frac{1}{2}, x\right)$

$$\frac{1}{6}\pi \tag{1.4.1}$$

Hvis du vil have samtlige løsninger returneret, skal du sætte variablen *_EnvAllSolutions* til *true* :

> $\_EnvAllSolutions := true$

$$\_EnvAllSolutions := true \tag{1.4.2}$$

> $FL := solve\left(\sin(x)=\frac{1}{2}, x\right)$

(1.4.3)

$$FL := \frac{1}{6}\pi + \frac{2}{3}\pi\_B1\sim + 2\pi\_Z1\sim \quad \textbf{(1.4.3)}$$

Resultatet skal forstås på den måde, at *_Z1~* er et vilkårligt helt tal, og variablen *_B~* kan kun antage værdierne 0 og 1. Tegnet ~ (tilde) efter _Z betyder, at Maple har gjort antagelser om *_Z1* og *_B1* (hvis du genberegner skifter navnene til *_Z2* og *_B2*).
Variablerne *_B1~* og *_Z1~* er lokale variabler. Dette har den konsekvens, at du ikke har direkte adgang til at tildele dem værdier, men de kan dog fiskes ud fra *FL* ved brug af *indets* (= indeterminates):

> $indets(FL)$

$$\{\_B1\sim, \_Z1\sim\} \quad \textbf{(1.4.4)}$$

Mængdeklammerne fjernes ved at bede om operanderne i udtrykket:

> $op(\textbf{(1.4.4)})$

$$\_B1\sim, \_Z1\sim \quad \textbf{(1.4.5)}$$

Naturligvis kan dette klares i én arbejdsgang, hvor vi desuden navngiver den fremkomne liste af 'ubestemte' med *z*. Dette sker for lette referencen til de 'ubestemte' (det er meget besværligt at få skrevet *_Z1~*):

> $z := op(indets(FL))$

$$z := \_B1\sim, \_Z1\sim \quad \textbf{(1.4.6)}$$

> $about(z_1)$

```
Originally _B1, renamed _B1~:
  is assumed to be: OrProp(0,1)
```

> $about(z_2)$

```
Originally _Z1, renamed _Z1~:
  is assumed to be: integer
```

En specifik løsning findes således

> $subs(\{z_1 = 1, z_2 = 0\}, FL)$

$$\frac{5}{6}\pi \quad \textbf{(1.4.7)}$$

Lad os lige stille *_EnvAllSolutions* tilbage og rense variablen z:

> $\_EnvAllSolutions :=$'$\_EnvAllSolutions$':
> $z :=$'z':

Ingen løsning

Har en ligning ingen løsninger, eller kan Maple ikke finde nogen, så returneres ingenting: Mere præcist den tomme udtryksfølge returneres (NULL).

> $solve(e^x = 0, x)$

Den næste ligning har løsning, men Maple kan ikke finde en formel for den:

> $solve(\sin(x) = x-1, x)$

$$RootOf(\_Z - \sin(\_Z) - 1) \tag{1.5.1}$$

Ved at sætte informationsniveauet for *solve* til 1 fås en advarsel, når *solve* ikke kan finde en løsning:

> $infolevel[solve] := 1 :$

> $solve(e^x = 0, x)$

```
solve: Warning: no solutions found
```

Sæt infolevel for solve tilbage til 0

> $infolevel[solve] := 0 :$

Numerisk løsning

Vi betragter nu en ligning, om hvilken det vides, at løsningerne ikke alle kan udtrykkes ved de til rådighed værende funktionsudtryk:

> $eq3 := 2 \cdot \sin(x) = x$

$$eq3 := 2\sin(x) = x \tag{1.6.1}$$

> $solve(eq3, x)$

$$RootOf(\_Z - 2\sin(\_Z)) \tag{1.6.2}$$

Vi fik et svar, men dette ligner nærmest en omformulering af spørgsmålet. Vi må derfor løse ligningen numerisk.
Der er to former fuldstændig som for *solve*.

> $fsolve(eq3, x)$

$$0. \tag{1.6.3}$$

> $fsolve(eq3, \{x\})$

$$\{x = 0.\} \tag{1.6.4}$$

Det bemærkes, at vi kun får én løsning. De andre kan fås ved enten at angive et søgeinterval eller et startgæt. Igen kan de to forskellige former med og uden { } bruges. Først prøver vi at angive et søgeinterval:

> $fsolve(eq3, x = 1..2)$

$$1.895494267 \tag{1.6.5}$$

> $fsolve(eq3, \{x = -2..1\})$

$$\{x = -1.895494267\} \tag{1.6.6}$$

Her prøver vi med et startgæt:

> $fsolve(eq3, x = 2)$

$$1.895494267 \tag{1.6.7}$$

> $fsolve(eq3, \{x = -3\})$

$$\{x = -1.895494267\} \tag{1.6.8}$$

Man kan fortælle Maple, at den skal undgå en løsning, fx hvis der (som her) er tale om en triviel løsning, som vi ikke behøver Maple til at finde:

> $fsolve(eq3, x, avoid = \{x = 0\})$

$$-1.895494267 \tag{1.6.9}$$

fsolve returnerer uevalueret, når den ikke kan finde en løsning:

> $fsolve(eq3, x = 2..3)$

$$fsolve(2\sin(x) = x, x, 2..3) \tag{1.6.10}$$

Det er en god ide at plotte i forbindelse med løsning af ligninger, fx kunne man plotte venstre- og højresiden af *eq2*. Maple kommandoerne rhs og lhs udtrækker hhv. høje- og venstre side af en ligning:

> $plot(\{lhs(eq3), rhs(eq3)\}, x = -3..3)$



fsolve accepterer følgende varianter med søgeinterval foruden de tilsvarende med krøllede parenteser:

> $fsolve(\sin, 4..7)$

$$6.283185307 \tag{1.6.11}$$

> $fsolve(\sin(x), x = 4..7)$

$$6.283185307 \tag{1.6.12}$$

> $fsolve(\sin(x) = 0, x = 4..7)$

$$6.283185307 \tag{1.6.13}$$

> $fsolve(\sin(x), x, 4..7)$ #4 ..7 *kan erstattes af* $x = 4..7$

$$6.283185307 \tag{1.6.14}$$

Lambert kommer på banen

Maples evne til at løse ligninger er ganske omfattende, og der kan sagtens dukke overraskelser op i form af funktioner, man aldrig har hørt om. Lad os se på ligningen

$$x^2 = 2^x.$$

Almindelig folklore siger, at denne ligning indeholder transcendente funktioner (2^x) på en ikke-algebraisk måde, og at sammenblandingen af algebraiske funktioner (x^2) og transcendente funktioner (2^x) ikke tillader en symbolsk løsning. Traditionelt vil man derfor løse ligningen numerisk (vha *fsolve*).

Grafen afslører, at der er 3 reelle løsninger til ligningen

> $plot(\{x^2, 2^x\}, x = -5..5, y = -5..20, scaling = unconstrained)$



Der er naturligvis de to ret trivielle løsninger $x = 2$ og $x = 4$, og en tredje mellem -1 og 0:

> $fsolve(x^2 = 2^x, x = -1..0)$

$$-0.7666646960 \quad \textbf{(1.7.1)}$$

Men symbolsk regning kan også lade sig gøre

> $solve(x^2 = 2^x, x)$

$$2, 4, -\frac{2\,\mathrm{LambertW}\left(\frac{1}{2}\ln(2)\right)}{\ln(2)} \quad \textbf{(1.7.2)}$$

Helt fint, men hvad er LambertW for en størrrelse?
I følge Maples hjælpetekster er LambertW den omvendte funktion til $y = x \cdot e^x$. Et blik på grafen for $y = x \cdot e^x$ afslører, at funktionen ikke er injektiv



Minimum befinder sig i $x = -1$, og minimumsværdien er $-e^{-1}$. Når vi laver den omvendte, er vi derfor nødt til at dele op i to: Grenen gennem $(0, 0)$, der ved omvending giver LambertW-funktionen, og den anden gren (i 3. kvadrant), der ved omvending giver den såkaldte LambertW (-1,x) funktion. Definitions- og værdimængder for de to funktioner er således:

$\mathrm{LambertW} \quad : [-e^{-1}, \infty[\;\; \rightarrow \;\; [-1, \infty[$

$\mathrm{LambertW}_{-1} : [-e^{-1}, 0[\;\; \rightarrow \;\; [-\infty, -1[$

Det betyder, at ligningen $y \cdot e^{y} = x$ har løsningen

$$y = \mathrm{LambertW}(x) \quad \text{hvis} \quad x \geq 0$$

og løsningerne

$$y = \mathrm{LambertW}(x) \;\; \vee \;\; y = \mathit{LambertW}(-1, x) \quad \text{hvis} \quad -e^{-1} \leq x \leq 0.$$

Hvis du vil studere LambertW funktionerne nærmere, kan du få Maples procedure til beregning skrevet ud i detaljer. Gør således fjern kolon efter $\mathit{eval}(\mathrm{LambertW})$ inden du udfører kommandoen.

> $interface(verboseproc = 3)$:
> $eval(\text{LambertW})$:

Flere ligninger

Skal du løse et system bestående af flere ligninger med flere ubekendte, skal du blot bundte ligningerne med {} og tilsvarende med variablerne. Fx

> $solve\left(\left\{x + 2y = 3, y + \frac{1}{x} = 1\right\}, \{x, y\}\right)$

$$\{x = -1, y = 2\}, \left\{x = 2, y = \frac{1}{2}\right\} \quad \textbf{(1.8.1)}$$

Den numeriske ligningsløser *fsolve* løser også ligningssystemer. Der er tre varianter af syntaksen:

> $eq4 := \{2x + y = 0, 3x^2 - y^4 = 0\}$

$$eq4 := \{2\,x + y = 0, 3\,x^2 - y^4 = 0\} \quad \textbf{(1.8.2)}$$

> $fsolve(eq4, \{x, y\})$

$$\{x = 0.4330127019, y = -0.8660254038\} \quad \textbf{(1.8.3)}$$

Med startgæt:

> $fsolve(eq4, \{x = 0, y = 1\})$

$$\{x = -0.4330127019, y = 0.8660254038\} \quad \textbf{(1.8.4)}$$

Med søgeintervaller

> $fsolve(eq4, \{x = 0 ..1, y = -2 ..0\})$

$$\{x = 0.4330127019, y = -0.8660254038\} \quad \textbf{(1.8.5)}$$

Det er naturligvis en god idé at bruge implicitplot først, så man kan se, hvor der er skæringspunkter:

> $plots[implicitplot](eq4, x = -1 ..1, y = -1 ..1, numpoints = 10000)$



Implicitte løsninger

Til tider vil du se løsninger returneret implicit i termer af *RootOf( )*, som fx i ligningssystemet:

> $solve(\{x^2 \cdot y^2 - 2 = 0, x^2 + y^2 - 4 = 0\}, [x, y])$

$$[[x = RootOf(\_Z^2 - 4 + RootOf(-4\_Z^2 + \_Z^4 + 2)^2), y = RootOf(-4\_Z^2 + \_Z^4 + 2)]] \quad \textbf{(1.9.1)}$$

For at få alle løsninger frem, benyttes kommandoen allvalues:

> $allvalues(\textbf{(1.9.1)})$

$$\left[\left[x = \sqrt{2+\sqrt{2}}, y = \sqrt{2-\sqrt{2}}\right]\right], \left[\left[x = -\sqrt{2+\sqrt{2}}, y = \sqrt{2-\sqrt{2}}\right]\right], \left[\left[x = \sqrt{2-\sqrt{2}}, y = \sqrt{2+\sqrt{2}}\right]\right], \left[\left[x = -\sqrt{2-\sqrt{2}}, y = \sqrt{2+\sqrt{2}}\right]\right], \left[\left[x = \sqrt{2+\sqrt{2}}, y = -\sqrt{2-\sqrt{2}}\right]\right], \left[\left[x = -\sqrt{2+\sqrt{2}}, y = -\sqrt{2-\sqrt{2}}\right]\right], \left[\left[x = \sqrt{2-\sqrt{2}}, y = -\sqrt{2+\sqrt{2}}\right]\right], \left[\left[x = -\sqrt{2-\sqrt{2}}, y = -\sqrt{2+\sqrt{2}}\right]\right] \quad \textbf{(1.9.2)}$$

Hvis du vil undgå alt det *RootOf( )* stads, kan du sætte variablen *_EnvExplicit* til *true*.

Også her er det en god ide at checke grafisk, at alle løsninger er med:

> $plots[implicitplot](\{x^2 \cdot y^2 - 2 = 0, x^2 + y^2 - 4 = 0\}, x = -5..5, y = -5..5, numpoints = 2000)$



Uligheder

Vi har allerede set, hvordan vi kan benytte en ulighed til at begrænse løsningsmængden. Altså fx

> $eq5 := 3x^3 + x^2 + x + 3 = 0$

$$eq5 := 3x^3 + x^2 + x + 3 = 0 \tag{1.10.1}$$

> $solve(eq5, x)$

$$-1, \frac{1}{3} + \frac{2}{3} \mathrm{I}\sqrt{2}, \frac{1}{3} - \frac{2}{3} \mathrm{I}\sqrt{2} \tag{1.10.2}$$

> $solve(\{eq5, x < 0\}, x)$

$$\{x = -1\} \tag{1.10.3}$$

Almindelige uligheder håndteres således:

> $solve(x^2 - 3x + 2 > 0, x)$

$$RealRange(-\infty, Open(1)), RealRange(Open(2), \infty) \quad \textbf{(1.10.4)}$$

Hvor resultatet skal tolkes således: $L =]-\infty, 1[\cup]2, \infty[$.

> $solve(x^2 - 3x + 2 < 0, x)$

$$RealRange(Open(1), Open(2)) \quad \textbf{(1.10.5)}$$

Manipulation af ligninger

Ligninger kan navngives og manipuleres som andre objekter. Maple understøtter lineære regninger med ligninger, dvs., de kan lægges sammen og trækkes fra hinanden ligesom man kan gange og dividere ligninger med skalarer.
Ligningssystemet

$$\begin{cases} 2x + 3y = 8 \\ 4x - y = 2 \end{cases}$$

kan løses i fuld detalje med de lineære operationer

> $eq1 := 2x + 3y = 8 :$
> $eq2 := 4x - y = 2 :$

> $eq3 := 2 \cdot eq1 - eq2$

$$eq3 := 7\,y = 14 \quad \textbf{(1.11.1)}$$

> $y = solve(eq3, y)$

$$y = 2 \quad \textbf{(1.11.2)}$$

> $eq4 := eq1 + 3 \cdot eq2$

$$eq4 := 14\,x = 14 \quad \textbf{(1.11.3)}$$

> $x = solve(eq4, x)$

$$x = 1 \quad \textbf{(1.11.4)}$$

Alternativt kan den sidste del løses ved at substituere $y = 2$ i fx *eq2*, og løse mht. x :

> $eq4 := subs(y = 2, eq2)$

$$eq4 := 4\,x - 2 = 2 \quad \textbf{(1.11.5)}$$

> $x = solve(eq4, x)$

$$x = 1 \quad \textbf{(1.11.6)}$$

Naturligvis kan ligningssystemet også løses i ét hug

> $solve(\{eq1, eq2\}, \{x, y\})$

$$\{x = 1, y = 2\} \quad \textbf{(1.11.7)}$$

Ved mere komplicerede operationer er der et par gode teknikker: Venstresiden og højresiden af ligningen kan udskilles vha kommandoerne

> *lhs(Ligning)*
> *rhs(Ligning)*

hvor *lhs* står for left-hand-side og *rhs* for right-hand-side. Herefter kan den samme operation på begge sider af lighedstegnet. Hvis du fx vil kvadrere, skrives det således

> $eqSq := lhs(eq)^2 = rhs(eq)^2$

Den anden teknik er en generel Maple teknik, hvor den universelle operator *map* benyttes til at udføre samme operation på begge sider af lighedstegnet. Hvis du fx vil kvadrere, skrives det således

> $eqSq := map\left(z \rightarrow z^2, eq\right)$

Nedenfor gives et par eksempler på de to teknikker til løsning af ligningen

$$\sqrt{x-2} = x-4$$

Først med lhs og rhs:

> $eq := \sqrt{x-2} = x-4$

$$eq := \sqrt{x-2} = x-4 \quad \textbf{(1.11.8)}$$

> $eq1 := lhs(eq)^2 = rhs(eq)^2$

$$eq1 := x-2 = (x-4)^2 \quad \textbf{(1.11.9)}$$

> $eq2 := lhs(eq1) - rhs(eq1) = 0$

$$eq2 := x-2-(x-4)^2 = 0 \quad \textbf{(1.11.10)}$$

> $eq3 := expand(eq2)$

$$eq3 := 9\,x - 18 - x^2 = 0 \quad \textbf{(1.11.11)}$$

> $L := solve(eq3, \{x\})$

$$L := \{x = 3\}, \{x = 6\} \quad \textbf{(1.11.12)}$$

Dernæst med map

> $eq := \sqrt{x-2} = x-4$

$$eq := \sqrt{x-2} = x-4 \quad \textbf{(1.11.13)}$$

> $eq1 := map\left(z \rightarrow z^2, eq\right)$

$$eq1 := x-2 = (x-4)^2 \quad \textbf{(1.11.14)}$$

etc.

Da vi selv har pillet ved ligningerne, kan der være sluppet fremmedrødder ind i ligningen. Det kan undersøges ved at sætte løsningerne ind i den oprindelige ligning:

> $subs\left(L_1, eq\right)$

$$1 = -1 \quad \textbf{(1.11.15)}$$

> $subs\left(L_2, eq\right)$

$$\sqrt{4} = 2 \quad \textbf{(1.11.16)}$$

> $simplify(\textbf{(1.11.16)})$

$$2 = 2 \tag{1.11.17}$$

Øvelser

Løs ligningssystemet, som gælder for vilkårlige trekanter (arealformel og cosinusrelationen), men ikke med hensyn til noget bestemt - dvs. $solve(\{eq1, eq2\})$:

> $eq1 := T = \frac{1}{2} a \cdot b \cdot \sin(C)$:

> $eq2 := c^2 = a^2 + b^2 - 2 \cdot a \cdot b \cdot \cos(C)$:

Maple leder nu efter alle mulige relationer mellem de indgående størrelser. Den finder to almene relationer. I det ene - ikke trivielle - tilfælde slipper Maple af med arealet T, og finder herved den omvendte cosinusrelation. I det andet - ikke trivielle - tilfælde slipper Maple af med C, og finder herved en version af Herons formel.

Kvadrer denne løsning, og faktoriser det fremkomne resultat med factor. Hvad skal der til før Herons formel står der?